

An Automated Linear Least Squares Solution Generator

Marc A. Murison

Astronomical Applications Department
U.S. Naval Observatory
Washington, DC

murison@riemann.usno.navy.mil
<http://aa.usno.navy.mil/murison/>

26 May, 1999

1. Introduction

In day-to-day work, one often has need to perform a quick linear least squares calculation for some model which approximates a physical process. Since the model is likely to be something other than simple linear regression, one winds up (often re-) deriving the appropriate normal equations and solving for the model parameters. This is wasted effort, thanks to the availability of computer algebra systems, in which it is easy to automate this process. In the section which follows, I illustrate the basic "quick and dirty" linear least squares process by means of a simple example. In section 3, I present a Maple procedure that performs the algebra automatically for any model that is linear in the parameters, and in section 4 are a few examples of its usage.

2. A Linear Least Squares Example

Start with a simple second degree polynomial as our model: $model := A + B t_i + C t_i^2$. Then the χ^2 statistic is $\chi^2 = \sum_{i=1}^N (y_i - model)^2$.

$$\chi^2 = \sum_{i=1}^N \left(y_i - A - B t_i - C t_i^2 \right)^2$$

where the sum is over N available data points, and the y_i are the data values, measured at times t_i . In keeping with the "quick and dirty" premise, we do not take into account *a priori* any weighting of the data. It is a simple matter to add weight factors into the solutions afterwards; including them now would serve only to clutter up the notation.

Next, we calculate the partial derivatives of χ^2 with respect to the model parameters, setting these derivatives to zero and thus forming the normal equations.

$$\frac{\partial}{\partial A} \text{rhs}(\%) = 0$$

$$\frac{\partial}{\partial B} \text{rhs}(%%) = 0$$

$$\frac{\partial}{\partial C} \text{rhs}(%%%%) = 0$$

$$\sum_{i=1}^N \left(-2y_i + 2A + 2Bt_i + 2Ct_i^2 \right) = 0$$

$$\sum_{i=1}^N \left(-2 \left(y_i - A - Bt_i - Ct_i^2 \right) t_i \right) = 0$$

$$\sum_{i=1}^N \left(-2 \left(y_i - A - Bt_i - Ct_i^2 \right) t_i^2 \right) = 0$$

NormalEqs := { %, %% , %%%% }

Finally, we solve the normal equations for the parameter values. These values are the linear least squares solution.

sols := sortsols(solve(value(NormalEqs), {A, B, C}), [A, B, C])

for *p* **in** *sols* **do** *print(p)* **od**

$$A = \left(\left(\sum_{i=1}^N y_i \right) \left(\sum_{i=1}^N t_i^3 \right)^2 - \left(\sum_{i=1}^N y_i \right) \left(\sum_{i=1}^N t_i^4 \right) \left(\sum_{i=1}^N t_i^2 \right) + \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i y_i \right) \left(\sum_{i=1}^N t_i^4 \right) \right.$$

$$- \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^2 y_i \right) - \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i y_i \right)$$

$$+ \left(\sum_{i=1}^N t_i^2 \right)^2 \left(\sum_{i=1}^N t_i^2 y_i \right) \right) / \left(-2 \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right) + N \left(\sum_{i=1}^N t_i^3 \right)^2 \right)$$

$$+ \left(\sum_{i=1}^N t_i^2 \right)^3 + \left(\sum_{i=1}^N t_i^4 \right) \left(\sum_{i=1}^N t_i^2 \right)^2 - \left(\sum_{i=1}^N t_i^4 \right) N \left(\sum_{i=1}^N t_i^2 \right) \right)$$

$$B = \left(\left(\sum_{i=1}^N t_i^4 \right) \left(\sum_{i=1}^N y_i \right) \left(\sum_{i=1}^N t_i \right) - \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i^2 y_i \right) \left(\sum_{i=1}^N t_i^2 \right) \right)$$

$$- \left(\sum_{i=1}^N t_i y_i \right) N \left(\sum_{i=1}^N t_i^4 \right) - \left(\sum_{i=1}^N y_i \right) \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right) + N \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^2 y_i \right) \right)$$

$$\begin{aligned}
& + \left(\sum_{i=1}^N t_i^2 \right)^2 \left(\sum_{i=1}^N t_i y_i \right) \Bigg) \Bigg/ \left(-2 \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right) + N \left(\sum_{i=1}^N t_i^3 \right)^2 \right. \\
& \left. + \left(\sum_{i=1}^N t_i^2 \right)^3 + \left(\sum_{i=1}^N t_i^4 \right) \left(\sum_{i=1}^N t_i^2 \right)^2 - \left(\sum_{i=1}^N t_i^4 \right) N \left(\sum_{i=1}^N t_i^2 \right) \right) \\
C = & \left(\left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i y_i \right) + N \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i y_i \right) \right. \\
& - \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N y_i \right) \left(\sum_{i=1}^N t_i \right) + \left(\sum_{i=1}^N t_i^2 y_i \right) \left(\sum_{i=1}^N t_i \right)^2 - \left(\sum_{i=1}^N t_i^2 y_i \right) N \left(\sum_{i=1}^N t_i^2 \right) \\
& \left. + \left(\sum_{i=1}^N t_i^2 \right)^2 \left(\sum_{i=1}^N y_i \right) \right) \Bigg/ \left(-2 \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right) + N \left(\sum_{i=1}^N t_i^3 \right)^2 \right. \\
& \left. + \left(\sum_{i=1}^N t_i^2 \right)^3 + \left(\sum_{i=1}^N t_i^4 \right) \left(\sum_{i=1}^N t_i^2 \right)^2 - \left(\sum_{i=1}^N t_i^4 \right) N \left(\sum_{i=1}^N t_i^2 \right) \right)
\end{aligned}$$

This is a bit easier to look at if we make a few substitutions.

$$\begin{aligned}
\text{subs} & \left(\text{seq} \left(\sum_{i=1}^N t_i^k = T_k, k=1..4 \right), \text{seq} \left(\sum_{i=1}^N y_i t_i^k = S_k, k=0..2 \right), \text{sols} \right) \\
A = & \frac{S_0 T_3^2 - S_0 T_4 T_2 + T_1 S_1 T_4 - T_1 T_3 S_2 - T_2 T_3 S_1 + T_2^2 S_2}{-2 T_1 T_2 T_3 + N T_3^2 + T_2^3 + T_4 T_1^2 - T_4 N T_2}, \\
B = & \frac{T_4 S_0 T_1 - T_1 S_2 T_2 - S_1 N T_4 - S_0 T_2 T_3 + N T_3 S_2 + T_2^2 S_1}{-2 T_1 T_2 T_3 + N T_3^2 + T_2^3 + T_4 T_1^2 - T_4 N T_2}, \\
C = & \frac{-T_1 T_2 S_1 + N T_3 S_1 - T_3 S_0 T_1 + S_2 T_1^2 - S_2 N T_2 + T_2^2 S_0}{-2 T_1 T_2 T_3 + N T_3^2 + T_2^3 + T_4 T_1^2 - T_4 N T_2}
\end{aligned}$$

Collecting on the mixed terms S_k , we have the result

`collect(%, [seq(S_k , $k=0..2$), N], factor)`

for p **in** % **do** `print(p)` **od**

$$\begin{aligned}
A &= \frac{\left(T_3^2 - T_4 T_2\right) S_0}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
&\quad + \frac{(-T_3 T_2 + T_1 T_4) S_1}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
&\quad + \frac{\left(T_2^2 - T_1 T_3\right) S_2}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
B &= \frac{(-T_3 T_2 + T_1 T_4) S_0}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
&\quad + \frac{(-N T_4 + T_2^2) S_1}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
&\quad + \frac{(N T_3 - T_1 T_2) S_2}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
C &= \frac{\left(T_2^2 - T_1 T_3\right) S_0}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
&\quad + \frac{(N T_3 - T_1 T_2) S_1}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2} \\
&\quad + \frac{\left(T_1^2 - N T_2\right) S_2}{\left(T_3^2 - T_4 T_2\right) N - 2 T_1 T_2 T_3 + T_2^3 + T_4 T_1^2}
\end{aligned}$$

3. An Automated Linear Least Squares Solution Generator

Here is a procedure that automates the process outlined in the example of the previous section.

```

# model must be of the form Y = F(t[k],P), where P are parameters
# and t is the independent variable. The sumindex subscript [k] MUST
# be attached to t in the model equation. For example,
#   leastsqrs( y=a+b*x[k], x, k, [a,b] );
# Also valid is
#   leastsqrs( y[k]=a+b*x[k], x, k, [a,b] );
#-----
leastsqrs := proc( model::`=`, indepvar::name,
                    sumindex::name, params::list )

local chi2, y, i, t, partials, k, eqs, solset, T, Y,
      replace_denominators, sums, YTsums, n, p, nmax, N,
      Tcount, Scount, ispoly, Delta, delta, tmp, goodsols;
global _subslist, time0;

if not has(model,sumindex) then
  ERROR("model must be of the form Y=F(t[i],P)");
fi;

time0 := time();

i := sumindex;
t := indepvar;
if not type(t,indexed) then
  t := t[i];
fi;
y := lhs(model);
if not type(y,indexed) then
  y := y[i];
fi;
chi2 := Sum((y-rhs(model))^2,i=1..N);
if type(rhs(model),polynom) then
  ispoly := true;
else
  ispoly := false;
fi;
debug_print(procname,"i,t,y,ispoly,chi2",4,i,t,y,ispoly,chi2);

#-----
# calculate the partials
#-----
debug_print(procname,"Forming the normal equations...",3);
partials := [];
for k from 1 to nops(params) do
  partials := [ op(partials), diff(chi2,params[k])=0 ];
od;
if printlevel >= 4 then
  debug_print(procname,"partials",4);
  for p in partials do
    print(p);
  od;
fi;

#-----
# form the summation substitutions variables
#-----
sums := convert( select( (x,y)->op(0,x)=y,

```

```

            indets(value(expand(partials)),function), sum ),
list );
#degree() can't handle summation subscripts, so remove them
YTsums    := subs( t=T, y=Y, sums );
_subsslist := [ ];
nmax      := 0;
debug_print(procname,"summation terms",4,sums);
debug_print(procname,"YT terms",5,YTsums);
Tcount   := 0;
Scount   := 0;
for k from 1 to nops(YTsums) do
  if ispoly then
    select(has,[op(YTsums[k])],T);
    if %=[ ] then
      n := 0;
    else
      n := degree( op(%), T );
    fi;
    nmax := max(n,nmax);
  fi;
  if has(YTsums[k],Y) then
    if not ispoly then
      n := Scount;
      nmax := max(n,nmax);
      Scount := Scount + 1;
    fi;
    _subsslist := [op(_subsslist),sums[k]='S'[n]];
  else
    if not ispoly then
      n := Tcount;
      Tcount := Tcount + 1;
    fi;
    _subsslist := [op(_subsslist),sums[k]='T'[n]];
  fi;
od;

#-----
# substitute for the summation terms in the partials
#-----
partials := collect( eval( value(expand(partials)), _subsslist ),
[seq(S[k],k=1..nmax),N], factor );

#put substitution list into form suitable for putting the sums back
_subsslist := map( (x)->rhs(x)=lhs(x), _subsslist );

#-----
# form the normal equations
#-----
eqs := collect( partials, [op(params),sum], factor );
if printlevel >= 1 then
  debug_print(procname,"normal equations",1);
  for p in eqs do
    print(p);
  od;
fi;

#-----

```

```

# solve the normal equations
#-----
debug_print(procname,"Solving the normal equations...",1);
_EnvAllSolutions := true;
map( allvalues, [solve( convert(eqs,set), convert(params,set) )] );
map( sortsols, %, params );
if nops(%)=1 then
    solset := op(%);
else
    solset := %;
    debug_print(procname,"There are ".(nops(solset))." solutions",1);
fi;
debug_print(procname,"raw solution(s)",4,solset);

#-----
# replace the denominators with another substitution
#-----

#first define a workhorse procedure
replace_denominators := proc( expr, N, nmax, S, Delta )

local k, p, d;
global _subslist;

#-----
# determine the common denominator
#-----
#grab and filter denominators from solution parameter expressions
map( denom, { seq(op(rhs(expr[k])),k=1..nops(expr)) } );
map( (x)->if type(x,{integer,fraction}) then 0 else x fi, % );
d := % minus {0};
if nops(d) > 1 then
    debug_print(procname,"denominator set",0,d);
    ERROR("Unable to determine a unique denominator!");
elif d={} then
    debug_print(procname,"no denominator for solution",2,expr);
    RETURN(expr);
else
    d := op(d);
fi;

#-----
# make the substitution in the solution
#-----
_subslist := [ Delta=eval(d,_subslist), op(_subslist) ];
collect( algsubs(d=Delta,expr), [Delta,seq(S[k],k=0..nmax),N], factor );
end;

#now make the substitutions
if type(solset,list(list)) then      #multiple solutions
for k from 1 to nops(solset) do
    subsop(k=replace_denominators(solset[k],N,nmax,S,Delta),solset);
    solset     := subs( Delta=delta[k], % );
    _subslist := subs( Delta=delta[k], _subslist );
od;
else                                #only one solution
    solset := replace_denominators(solset,N,nmax,S,Delta);

```

```

fi;

debug_print(procname,"substitution list",2,_subslist);

-----
# check the solution(s)
-----
if type(solset,list(list)) then #multiple solutions
  debug_print(procname,"Verifying the ".(nops(solset))." solutions...",1);
  goodsols := [];
  for k from 1 to nops(solset) do
    tmp := expand(subs(_subslist,eval(eqs,solset[k])));
    if not type(tmp,list(0=0)) then
      debug_print(procname,"Solution ".k." is invalid!",0,solset[k]);
      debug_print(procname,"Solution ".k." substituted into normal eqs",0,eqs);
      debug_print(procname,"Throwing out solution ".k,0);
    else
      goodsols := [op(goodsols),solset[k]];
    fi;
  od;
  solset := goodsols;
  if nops(solset)=1 then
    solset := op(solset);
  fi;
else
  #single solution
  debug_print(procname,"Verifying the solution...",1);
  eqs := factor(expand(subs(_subslist,eval(eqs,solset))));
  if not type(eqs,list(0=0)) then
    debug_print(procname,"Solution substituted into normal eqs",0,eqs);
    ERROR("Invalid solution!");
  fi;
fi;

if time()-time0 > 10 then
  debug_print(procname,"Done!",1);
fi;
solset;
end:

```

4. Examples

Polynomial of degree 1

We start with simple linear regression to check the program. We set the procedure's "chattiness" to maximum verbosity. The solution appears as the last bracketed expression.

printlevel := 4

```

leastsqrs( $y = A + B t_i$ ,  $t, i, [A, B]$ )
leastsqrs[0]:  $i, t, y, \text{ispoly}, \text{chi2}$ 

```

$$i, t_i, y_i, \text{true}, \sum_{i=1}^N (y_i - A - B t_i)^2$$

```
leastsqrs[0]: Forming the normal equations...
leastsqrs[0]: partials
```

$$\sum_{i=1}^N (-2y_i + 2A + 2B t_i) = 0$$

$$\sum_{i=1}^N (-2(y_i - A - B t_i) t_i) = 0$$

```
leastsqrs[0]: summation terms
```

$$\left[\sum_{i=1}^N y_i, \sum_{i=1}^N t_i y_i, \sum_{i=1}^N t_i^2, \sum_{i=1}^N t_i^2 \right]$$

```
leastsqrs[0]: normal equations
```

$$-2S_0 + 2AN + 2BT_1 = 0$$

$$-2S_1 + 2AT_1 + 2BT_2 = 0$$

```
leastsqrs[0]: Solving the normal equations...
leastsqrs[0]: raw solution(s)
```

$$\left[A = \frac{S_0 T_2 - T_1 S_1}{-T_1^2 + T_2 N}, B = -\frac{-S_1 N + T_1 S_0}{-T_1^2 + T_2 N} \right]$$

```
leastsqrs[0]: substitution list
```

$$\Delta = \left(\sum_{i=1}^N t_i \right)^2 + \left(\sum_{i=1}^N t_i^2 \right) N, S_0 = \sum_{i=1}^N y_i, S_1 = \sum_{i=1}^N t_i y_i, T_1 = \sum_{i=1}^N t_i^2, T_2 = \sum_{i=1}^N t_i^2$$

```
leastsqrs[0]: Verifying the solution...
```

$$\left[A = \frac{S_0 T_2 - T_1 S_1}{\Delta}, B = \frac{S_1 N - T_1 S_0}{\Delta} \right]$$

The substitutions are stored in the global variable `_sublist`. Hence, the full solution is

```
eval(% , _sublist)
```

$$\left[A = \frac{\left(\sum_{i=1}^N y_i \right) \left(\sum_{i=1}^N t_i^2 \right) - \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N t_i y_i \right)}{\left(\sum_{i=1}^N t_i \right)^2 + \left(\sum_{i=1}^N t_i^2 \right) N}, B = \frac{\left(\sum_{i=1}^N t_i y_i \right) N - \left(\sum_{i=1}^N t_i \right) \left(\sum_{i=1}^N y_i \right)}{\left(\sum_{i=1}^N t_i \right)^2 + \left(\sum_{i=1}^N t_i^2 \right) N} \right]$$

We can convert the solution into an optimized Maple procedure. This procedure can then be used to evaluate the model with a set of data. It is easily saved to disk for conversion into a C or fortran procedure — also quite easy.

```
tmp := array(map(x → rhs(x), %))
```

```

foo := optimize(makeproc(tmp, parameters = [N, y, t]))
foo := proc(N, y, t)
local t8, t11, t4, t5, t1, tmp, t2;
tmp := array(1 .. 2);
t1 := sum(y[i], i = 1 .. N);
t2 := sum(t[i]^2, i = 1 .. N);
t4 := sum(t[i], i = 1 .. N);
t5 := sum(t[i]*y[i], i = 1 .. N);
t8 := t4^2;
t11 := 1 / (-t8 + t2*N);
tmp[1] := (t1*t2 - t4*t5)*t11;
tmp[2] := (t5*N - t4*t1)*t11;
tmp
end
writeto("d:/Maple/test.p")
print(foo)
writeto(terminal)

```

This is simple and straightforward in this example. Later on, we will see that the solutions can get rather unwieldy, but that the optimized procedures created from them are still relatively simple.

Polynomial of degree 2

```

printlevel := 2
leastsqrs $\left(Y = A + B t_i + C t_i^2, t, i, [A, B, C]\right)$ 
leastsqrs[0]: normal equations
 $-2 S_0 + 2 A N + 2 B T_1 + 2 C T_2 = 0$ 
 $-2 S_1 + 2 A T_1 + 2 B T_2 + 2 C T_3 = 0$ 
 $-2 S_2 + 2 A T_2 + 2 B T_3 + 2 C T_4 = 0$ 
leastsqrs[0]: Solving the normal equations...
leastsqrs[0]: substitution list

$$\Delta = -2 \left( \sum_{i=1}^N t_i \right) \left( \sum_{i=1}^N t_i^2 \right) \left( \sum_{i=1}^N t_i^3 \right) + \left( \sum_{i=1}^N t_i^3 \right)^2 N + \left( \sum_{i=1}^N t_i^2 \right)^3 - \left( \sum_{i=1}^N t_i^4 \right) N \left( \sum_{i=1}^N t_i^2 \right)$$


$$+ \left( \sum_{i=1}^N t_i^4 \right) \left( \sum_{i=1}^N t_i \right)^2, S_0 = \sum_{i=1}^N Y_i, T_1 = \sum_{i=1}^N t_i, T_2 = \sum_{i=1}^N t_i^2, S_1 = \sum_{i=1}^N t_i Y_i, T_3 = \sum_{i=1}^N t_i^3,$$


```

$$S_2 = \sum_{i=1}^N t_i^2 Y_i, T_4 = \sum_{i=1}^N t_i^4 \quad \left[\right]$$

leastsqrs[0]: Verifying the solution...

$$\begin{aligned} A &= \frac{\left(T_3^2 - T_4 T_2 \right) S_0 + (-T_2 T_3 + T_1 T_4) S_1 + \left(-T_3 T_1 + T_2^2 \right) S_2}{\Delta}, \\ B &= \frac{(-T_2 T_3 + T_1 T_4) S_0 + \left(T_2^2 - T_4 N \right) S_1 + (-T_1 T_2 + T_3 N) S_2}{\Delta}, \\ C &= \frac{\left(-T_3 T_1 + T_2^2 \right) S_0 + (-T_1 T_2 + T_3 N) S_1 + \left(-T_2 N + T_1^2 \right) S_2}{\Delta} \end{aligned}$$

```

tmp := array(map(x → rhs(x), eval(%,
subslist)))
optimize(makeproc(tmp, parameters = [N, Y, t]))

proc(N, Y, t)
local t7, t26, t31, t12, t16, t28, t17, t13, t21, t18, t36, t4, t10, tmp, t3, t1, t2;
tmp := array(1 .. 3);
t1 := sum(t[i]^3, i = 1 .. N);
t2 := t1^2;
t3 := sum(t[i]^4, i = 1 .. N);
t4 := sum(t[i]^2, i = 1 .. N);
t7 := sum(Y[i], i = 1 .. N);
t10 := sum(t[i], i = 1 .. N);
t12 := -t4*t1 + t10*t3;
t13 := sum(t[i]*Y[i], i = 1 .. N);
t16 := t4^2;
t17 := -t1*t10 + t16;
t18 := sum(t[i]^2*Y[i], i = 1 .. N);
t21 := t10*t4;
t26 := t3*N;
t28 := t10^2;
t31 := 1 / (-2*t21*t1 + t2*N + t16*t4 - t26*t4 + t3*t28);
tmp[1] := ((t2 - t3*t4)*t7 + t12*t13 + t17*t18)*t31;
t36 := -t21 + t1*N;
tmp[2] := (t12*t7 + (t16 - t26)*t13 + t36*t18)*t31;
tmp[3] := (t17*t7 + t36*t13 + (-t4*N + t28)*t18)*t31;
tmp

```

end

Polynomial of degree 3

In this example, we will not show the solution, since it is a bit unwieldy.

leastsqrs $\left(Y = A + B t_i^1 + C t_i^2 + D t_i^3, t, i, [A, B, C, D] \right)$
 leastsqrs[0]: normal equations

$$-2 S_0 + 2 A N + 2 B T_1 + 2 C T_2 + 2 D T_3 = 0$$

$$-2 S_1 + 2 A T_1 + 2 B T_2 + 2 C T_3 + 2 D T_4 = 0$$

$$-2 S_2 + 2 A T_2 + 2 B T_3 + 2 C T_4 + 2 D T_5 = 0$$

$$-2 S_3 + 2 A T_3 + 2 B T_4 + 2 C T_5 + 2 D T_6 = 0$$

leastsqrs[0]: Solving the normal equations...
 leastsqrs[1]: substitution list

$$\begin{aligned} \Delta = & \left(\sum_{i=1}^N t_i^6 \right) \left(\sum_{i=1}^N t_i^2 \right)^3 - 2 \left(\sum_{i=1}^N t_i^2 \right)^2 \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^5 \right) - \left(\sum_{i=1}^N t_i^2 \right)^2 \left(\sum_{i=1}^N t_i^4 \right)^2 \\ & + \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^5 \right)^2 N - \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^6 \right) \left(\sum_{i=1}^N t_i^4 \right) N \\ & + 2 \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^4 \right) \left(\sum_{i=1}^N t_i^5 \right) \left(\sum_{i=1}^N t_i^2 \right) + 3 \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right)^2 \left(\sum_{i=1}^N t_i^4 \right) \\ & - 2 \left(\sum_{i=1}^N t_i^2 \right) \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^6 \right) \left(\sum_{i=1}^N t_i^3 \right) + \left(\sum_{i=1}^N t_i^4 \right)^3 N + \left(\sum_{i=1}^N t_i^6 \right) \left(\sum_{i=1}^N t_i^3 \right)^2 N \\ & - 2 \left(\sum_{i=1}^N t_i^4 \right) N \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^5 \right) - 2 \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^4 \right)^2 \\ & + 2 \left(\sum_{i=1}^N t_i^3 \right) \left(\sum_{i=1}^N t_i^5 \right)^2 \left(\sum_{i=1}^N t_i^2 \right) + \left(\sum_{i=1}^N t_i^6 \right)^2 \left(\sum_{i=1}^N t_i^4 \right) - \left(\sum_{i=1}^N t_i^3 \right)^2 \left(\sum_{i=1}^N t_i^5 \right)^2 \\ & - \left(\sum_{i=1}^N t_i^3 \right)^4, T_6 = \sum_{i=1}^N t_i^6, T_5 = \sum_{i=1}^N t_i^5, S_0 = \sum_{i=1}^N Y_i, T_2 = \sum_{i=1}^N t_i^2, T_1 = \sum_{i=1}^N t_i, T_3 = \sum_{i=1}^N t_i^3, \\ & S_1 = \sum_{i=1}^N t_i Y_i, S_3 = \sum_{i=1}^N t_i^3 Y_i, T_4 = \sum_{i=1}^N t_i^4, S_2 = \sum_{i=1}^N t_i^2 Y_i \end{aligned}$$

leastsqrs[2]: Verifying the solution...

map(cost, %)

[21 additions + 50 multiplications + divisions + assignments,
 22 additions + 48 multiplications + divisions + assignments,
 22 additions + 48 multiplications + divisions + assignments,
 21 additions + 47 multiplications + divisions + assignments]

The following commands produce an optimized procedure, which we will not show here.

```
tmp := array(map(x → rhs(x), eval(%%, _subslist)))
optimize(makeproc(tmp, parameters = [N, Y, t]))
```

Sinusoid — Representation 1

printlevel := 1

leastsqrs($Y = A \cos(2\pi v t_i + \phi)$, $t, i, [A, \phi]$)

leastsqrs[0]: normal equations

$$(8 \cos(\phi)^2 T_0 - 8 \cos(\phi)^2 T_1 - 16 \cos(\phi) \sin(\phi) T_2 + 2 \cos(\phi)^2 N + 8 \cos(\phi) \sin(\phi) T_3 + 8 \sin(\phi)^2 T_4) A - 4 \cos(\phi) S_0 + 2 \cos(\phi) S_1 + 4 \sin(\phi) S_2 = 0$$

$$(-2 \cos(\phi) \sin(\phi) N - 8 \cos(\phi) \sin(\phi) T_0 + 4 \cos(\phi)^2 T_3 - 8 \cos(\phi)^2 T_2 + 8 \cos(\phi) \sin(\phi) T_4 + 8 \cos(\phi) \sin(\phi) T_1 - 4 \sin(\phi)^2 T_3 + 8 \sin(\phi)^2 T_2) A^2 + (-2 \sin(\phi) S_1 + 4 \cos(\phi) S_2 + 4 \sin(\phi) S_0) A = 0$$

leastsqrs[0]: Solving the normal equations...

leastsqrs[2]: There are 3 solutions

leastsqrs[4]: Verifying the 3 solutions...

leastsqrs[18]: Solution 2 is invalid!

$$\begin{aligned} A = & \frac{1}{2} \sqrt{\left(16 T_2 S_2 S_1 T_4 + 16 T_3 S_2 S_0 T_4 - 8 T_3 S_2 S_1 T_4 - 32 T_2 S_2 S_0 T_4 - 16 T_2 S_2^2 T_3\right. \\ & \left.- 16 S_0 T_4^2 S_1 + 16 T_2^2 S_2^2 - 4 T_2 S_1^2 T_3 - 4 T_3^2 S_0 S_1 + 16 T_2 S_0 T_3 S_1 + 8 S_2 T_1 T_3 S_1\right. \\ & \left.- 8 S_2 T_0 T_3 S_1 + 32 S_2 T_1 T_2 S_0 - 16 S_2 T_1 T_2 S_1 - 16 S_2 T_1 T_3 S_0 - 16 T_2^2 S_0 S_1\right. \\ & \left.- 2 S_2 N T_3 S_1 - 32 S_2 T_0 T_2 S_0 + 16 S_2 T_0 T_2 S_1 + 16 S_2 T_0 T_3 S_0 + 4 S_2 N T_2 S_1\right. \\ & \left.+ 4 S_2 N T_3 S_0 - 16 T_2 S_0^2 T_3 + 8 S_2^2 N T_0 - 8 S_2^2 N T_1 - 32 S_2^2 T_0 T_1 + S_2^2 N^2\right. \\ & \left.+ 16 S_2^2 T_0^2 + 16 S_2^2 T_1^2 + 16 T_2^2 S_0^2 + 4 T_2^2 S_1^2 + 4 T_3^2 S_0^2 + T_3^2 S_1^2 - 8 S_2 N T_2 S_0\right)} \end{aligned}$$

$$\begin{aligned}
& + 4 T_3^2 S_2^2 + 16 S_0^2 T_4^2 + 4 S_1^2 T_4^2 \Big) \Big/ \delta_2, \phi = \arctan \left(\right. \\
& - S_2 N - 4 S_2 T_0 + 4 S_2 T_1 + 4 T_2 S_0 - 2 T_2 S_1 - 2 T_3 S_0 + T_3 S_1 \Big) \Big/ \sqrt{16 T_2 S_2 S_1 T_4} \\
& + 16 T_3 S_2 S_0 T_4 - 8 T_3 S_2 S_1 T_4 - 32 T_2 S_2 S_0 T_4 - 16 T_2 S_2^2 T_3 - 16 S_0 T_4^2 S_1 + 16 T_2^2 S_2^2 \\
& - 4 T_2 S_1^2 T_3 - 4 T_3^2 S_0 S_1 + 16 T_2 S_0 T_3 S_1 + 8 S_2 T_1 T_3 S_1 - 8 S_2 T_0 T_3 S_1 \\
& + 32 S_2 T_1 T_2 S_0 - 16 S_2 T_1 T_2 S_1 - 16 S_2 T_1 T_3 S_0 - 16 T_2^2 S_0 S_1 - 2 S_2 N T_3 S_1 \\
& - 32 S_2 T_0 T_2 S_0 + 16 S_2 T_0 T_2 S_1 + 16 S_2 T_0 T_3 S_0 + 4 S_2 N T_2 S_1 + 4 S_2 N T_3 S_0 \\
& - 16 T_2 S_0^2 T_3 + 8 S_2^2 N T_0 - 8 S_2^2 N T_1 - 32 S_2^2 T_0 T_1 + S_2^2 N^2 + 16 S_2^2 T_0^2 + 16 S_2^2 T_1^2 \\
& + 16 T_2^2 S_0^2 + 4 T_2^2 S_1^2 + 4 T_3^2 S_0^2 + T_3^2 S_1^2 - 8 S_2 N T_2 S_0 + 4 T_3^2 S_2^2 + 16 S_0^2 T_4^2 \\
& \left. + 4 S_1^2 T_4^2 \right) 2 (-2 T_2 S_2 + T_3 S_2 + 2 S_0 T_4 - S_1 T_4) \Big/ \sqrt{16 T_2 S_2 S_1 T_4 + 16 T_3 S_2 S_0 T_4} \\
& - 8 T_3 S_2 S_1 T_4 - 32 T_2 S_2 S_0 T_4 - 16 T_2 S_2^2 T_3 - 16 S_0 T_4^2 S_1 + 16 T_2^2 S_2^2 - 4 T_2 S_1^2 T_3 \\
& - 4 T_3^2 S_0 S_1 + 16 T_2 S_0 T_3 S_1 + 8 S_2 T_1 T_3 S_1 - 8 S_2 T_0 T_3 S_1 + 32 S_2 T_1 T_2 S_0 \\
& - 16 S_2 T_1 T_2 S_1 - 16 S_2 T_1 T_3 S_0 - 16 T_2^2 S_0 S_1 - 2 S_2 N T_3 S_1 - 32 S_2 T_0 T_2 S_0 \\
& + 16 S_2 T_0 T_2 S_1 + 16 S_2 T_0 T_3 S_0 + 4 S_2 N T_2 S_1 + 4 S_2 N T_3 S_0 - 16 T_2 S_0^2 T_3 + 8 S_2^2 N T_0 \\
& - 8 S_2^2 N T_1 - 32 S_2^2 T_0 T_1 + S_2^2 N^2 + 16 S_2^2 T_0^2 + 16 S_2^2 T_1^2 + 16 T_2^2 S_0^2 + 4 T_2^2 S_1^2 \\
& \left. + 4 T_3^2 S_0^2 + T_3^2 S_1^2 - 8 S_2 N T_2 S_0 + 4 T_3^2 S_2^2 + 16 S_0^2 T_4^2 + 4 S_1^2 T_4^2 \right) \Big) + 2 \pi Z
\end{aligned}$$

leastsqrs[18]: Solution 2 substituted into normal eqs

$$\begin{aligned}
& [(8 \cos(\phi)^2 T_0 - 8 \cos(\phi)^2 T_1 - 16 \cos(\phi) \sin(\phi) T_2 + 2 \cos(\phi)^2 N + 8 \cos(\phi) \sin(\phi) T_3 \\
& + 8 \sin(\phi)^2 T_4) A - 4 \cos(\phi) S_0 + 2 \cos(\phi) S_1 + 4 \sin(\phi) S_2 = 0, (-2 \cos(\phi) \sin(\phi) N \\
& - 8 \cos(\phi) \sin(\phi) T_0 + 4 \cos(\phi)^2 T_3 - 8 \cos(\phi)^2 T_2 + 8 \cos(\phi) \sin(\phi) T_4 + 8 \cos(\phi) \sin(\phi) T_1 \\
& - 4 \sin(\phi)^2 T_3 + 8 \sin(\phi)^2 T_2) A^2 + (-2 \sin(\phi) S_1 + 4 \cos(\phi) S_2 + 4 \sin(\phi) S_0) A = 0]
\end{aligned}$$

leastsqrs[18]: Throwing out solution 2

leastsqrs[31]: Solution 3 is invalid!

$$\begin{aligned}
A = & -\frac{1}{2} \operatorname{sqrt} \left(16 T_2 S_2 S_1 T_4 + 16 T_3 S_2 S_0 T_4 - 8 T_3 S_2 S_1 T_4 - 32 T_2 S_2 S_0 T_4 - 16 T_2 S_2^2 T_3 \right. \\
& - 16 S_0 T_4^2 S_1 + 16 T_2^2 S_2^2 - 4 T_2 S_1^2 T_3 - 4 T_3^2 S_0 S_1 + 16 T_2 S_0 T_3 S_1 + 8 S_2 T_1 T_3 S_1 \\
& - 8 S_2 T_0 T_3 S_1 + 32 S_2 T_1 T_2 S_0 - 16 S_2 T_1 T_2 S_1 - 16 S_2 T_1 T_3 S_0 - 16 T_2^2 S_0 S_1 \\
& - 2 S_2 N T_3 S_1 - 32 S_2 T_0 T_2 S_0 + 16 S_2 T_0 T_2 S_1 + 16 S_2 T_0 T_3 S_0 + 4 S_2 N T_2 S_1 \\
& + 4 S_2 N T_3 S_0 - 16 T_2 S_0^2 T_3 + 8 S_2^2 N T_0 - 8 S_2^2 N T_1 - 32 S_2^2 T_0 T_1 + S_2^2 N^2 \\
& + 16 S_2^2 T_0^2 + 16 S_2^2 T_1^2 + 16 T_2^2 S_0^2 + 4 T_2^2 S_1^2 + 4 T_3^2 S_0^2 + T_3^2 S_1^2 - 8 S_2 N T_2 S_0 \\
& \left. + 4 T_3^2 S_2^2 + 16 S_0^2 T_4^2 + 4 S_1^2 T_4^2 \right) / \delta_3, \phi = \arctan \left(- \left(\right. \right. \\
& - S_2 N - 4 S_2 T_0 + 4 S_2 T_1 + 4 T_2 S_0 - 2 T_2 S_1 - 2 T_3 S_0 + T_3 S_1) / \operatorname{sqrt} \left(16 T_2 S_2 S_1 T_4 \right. \\
& + 16 T_3 S_2 S_0 T_4 - 8 T_3 S_2 S_1 T_4 - 32 T_2 S_2 S_0 T_4 - 16 T_2 S_2^2 T_3 - 16 S_0 T_4^2 S_1 + 16 T_2^2 S_2^2 \\
& - 4 T_2 S_1^2 T_3 - 4 T_3^2 S_0 S_1 + 16 T_2 S_0 T_3 S_1 + 8 S_2 T_1 T_3 S_1 - 8 S_2 T_0 T_3 S_1 \\
& + 32 S_2 T_1 T_2 S_0 - 16 S_2 T_1 T_2 S_1 - 16 S_2 T_1 T_3 S_0 - 16 T_2^2 S_0 S_1 - 2 S_2 N T_3 S_1 \\
& - 32 S_2 T_0 T_2 S_0 + 16 S_2 T_0 T_2 S_1 + 16 S_2 T_0 T_3 S_0 + 4 S_2 N T_2 S_1 + 4 S_2 N T_3 S_0 \\
& - 16 T_2 S_0^2 T_3 + 8 S_2^2 N T_0 - 8 S_2^2 N T_1 - 32 S_2^2 T_0 T_1 + S_2^2 N^2 + 16 S_2^2 T_0^2 + 16 S_2^2 T_1^2 \\
& + 16 T_2^2 S_0^2 + 4 T_2^2 S_1^2 + 4 T_3^2 S_0^2 + T_3^2 S_1^2 - 8 S_2 N T_2 S_0 + 4 T_3^2 S_2^2 + 16 S_0^2 T_4^2 \\
& \left. \left. + 4 S_1^2 T_4^2 \right) - 2 (-2 T_2 S_2 + T_3 S_2 + 2 S_0 T_4 - S_1 T_4) / \operatorname{sqrt} \left(16 T_2 S_2 S_1 T_4 \right. \right. \\
& + 16 T_3 S_2 S_0 T_4 - 8 T_3 S_2 S_1 T_4 - 32 T_2 S_2 S_0 T_4 - 16 T_2 S_2^2 T_3 - 16 S_0 T_4^2 S_1 + 16 T_2^2 S_2^2 \\
& - 4 T_2 S_1^2 T_3 - 4 T_3^2 S_0 S_1 + 16 T_2 S_0 T_3 S_1 + 8 S_2 T_1 T_3 S_1 - 8 S_2 T_0 T_3 S_1 \\
& + 32 S_2 T_1 T_2 S_0 - 16 S_2 T_1 T_2 S_1 - 16 S_2 T_1 T_3 S_0 - 16 T_2^2 S_0 S_1 - 2 S_2 N T_3 S_1 \\
& - 32 S_2 T_0 T_2 S_0 + 16 S_2 T_0 T_2 S_1 + 16 S_2 T_0 T_3 S_0 + 4 S_2 N T_2 S_1 + 4 S_2 N T_3 S_0 \\
& - 16 T_2 S_0^2 T_3 + 8 S_2^2 N T_0 - 8 S_2^2 N T_1 - 32 S_2^2 T_0 T_1 + S_2^2 N^2 + 16 S_2^2 T_0^2 + 16 S_2^2 T_1^2
\end{aligned}$$

$$+ 16 T_2^2 S_0^2 + 4 T_2^2 S_1^2 + 4 T_3^2 S_0^2 + T_3^2 S_1^2 - 8 S_2 N T_2 S_0 + 4 T_3^2 S_2^2 + 16 S_0^2 T_4^2 \\ + 4 S_1^2 T_4^2 \Big) \Big) + 2 \pi Z \Big]$$

leastsqrs[31]: Solution 3 substituted into normal eqs

$$[(8 \cos(\phi)^2 T_0 - 8 \cos(\phi)^2 T_1 - 16 \cos(\phi) \sin(\phi) T_2 + 2 \cos(\phi)^2 N + 8 \cos(\phi) \sin(\phi) T_3 \\ + 8 \sin(\phi)^2 T_4) A - 4 \cos(\phi) S_0 + 2 \cos(\phi) S_1 + 4 \sin(\phi) S_2 = 0, (-2 \cos(\phi) \sin(\phi) N \\ - 8 \cos(\phi) \sin(\phi) T_0 + 4 \cos(\phi)^2 T_3 - 8 \cos(\phi)^2 T_2 + 8 \cos(\phi) \sin(\phi) T_4 + 8 \cos(\phi) \sin(\phi) T_1 \\ - 4 \sin(\phi)^2 T_3 + 8 \sin(\phi)^2 T_2) A^2 + (-2 \sin(\phi) S_1 + 4 \cos(\phi) S_2 + 4 \sin(\phi) S_0) A = 0]$$

leastsqrs[31]: Throwing out solution 3

leastsqrs[31]: Done!

$$\left[A = 0, \phi = \arctan\left(\frac{1}{2} \frac{2 S_0 - S_1}{S_2}\right) + \pi Z \right]$$

As we can see, $Y = A \cos(2 \pi v t_i + \phi)$ is not a particularly good choice of representation, since the solution involves inverse trigonometric functions and is particularly useless ($A = 0$). The next, equivalent, example leads to a useful solution.

Sinusoid — Representation 2

printlevel := 2

leastsqrs($Y = A \cos(2 \pi v t_i) + B \sin(2 \pi v t_i)$, t , i , [A, B])

leastsqrs[0]: normal equations

$$(-8 T_1 + 2 N + 8 T_0) A + (-4 T_3 + 8 T_2) B - 4 S_2 + 2 S_0 = 0$$

$$(-4 T_3 + 8 T_2) A - 4 S_1 + 8 B T_4 = 0$$

leastsqrs[0]: Solving the normal equations...

leastsqrs[1]: substitution list

$$\left[\Delta = 4 \left(\sum_{i=1}^N \cos(\pi v t_i)^4 \right) \left(\sum_{i=1}^N \sin(\pi v t_i)^2 \cos(\pi v t_i)^2 \right) \right. \\ \left. - 4 \left(\sum_{i=1}^N \cos(\pi v t_i)^2 \right) \left(\sum_{i=1}^N \sin(\pi v t_i)^2 \cos(\pi v t_i)^2 \right) + N \left(\sum_{i=1}^N \sin(\pi v t_i)^2 \cos(\pi v t_i)^2 \right) \right. \\ \left. - 4 \left(\sum_{i=1}^N \cos(\pi v t_i)^3 \sin(\pi v t_i) \right)^2 \right]$$

$$\begin{aligned}
& + 4 \left(\sum_{i=1}^N \sin(\pi v t_i) \cos(\pi v t_i) \right) \left(\sum_{i=1}^N \cos(\pi v t_i)^3 \sin(\pi v t_i) \right) \\
& - \left(\sum_{i=1}^N \sin(\pi v t_i) \cos(\pi v t_i) \right)^2, S_0 = \sum_{i=1}^N Y_i, T_0 = \sum_{i=1}^N \cos(\pi v t_i)^4, T_1 = \sum_{i=1}^N \cos(\pi v t_i)^2, \\
& T_2 = \sum_{i=1}^N \cos(\pi v t_i)^3 \sin(\pi v t_i), T_3 = \sum_{i=1}^N \sin(\pi v t_i) \cos(\pi v t_i), \\
& S_1 = \sum_{i=1}^N Y_i \sin(\pi v t_i) \cos(\pi v t_i), T_4 = \sum_{i=1}^N \sin(\pi v t_i)^2 \cos(\pi v t_i)^2, S_2 = \sum_{i=1}^N Y_i \cos(\pi v t_i)^2 \\
& \quad \left. \right]
\end{aligned}$$

leastsqrs[1]: Verifying the solution...

$$\begin{aligned}
A &= \frac{-S_0 T_4 + (-2 T_2 + T_3) S_1 + 2 S_2 T_4}{\Delta}, \\
B &= \frac{\left(T_2 - \frac{1}{2} T_3 \right) S_0 + \left(2 T_0 - 2 T_1 + \frac{1}{2} N \right) S_1 + (-2 T_2 + T_3) S_2}{\Delta}
\end{aligned}$$

```

tmp := array(map(x → rhs(x), eval(%_, _subalist)))
optimize(makeproc(tmp, parameters = [N, Y, t]))

proc(N, Y, t)
local tmp, t1, t2, t8, t14, t7, t6, t4, t10, t17, t21, t27, t25;
tmp := array(1 .. 2);
t1 := sum(Y[i], i = 1 .. N);
t2 := sum(sin(pi*v*t[i])^2*cos(pi*v*t[i])^2, i = 1 .. N);
t4 := sum(cos(pi*v*t[i])^3*sin(pi*v*t[i]), i = 1 .. N);
t6 := sum(sin(pi*v*t[i])*cos(pi*v*t[i]), i = 1 .. N);
t7 := -2*t4 + t6;
t8 := sum(Y[i]*sin(pi*v*t[i])*cos(pi*v*t[i]), i = 1 .. N);
t10 := sum(Y[i]*cos(pi*v*t[i])^2, i = 1 .. N);
t14 := sum(cos(pi*v*t[i])^4, i = 1 .. N);
t17 := sum(cos(pi*v*t[i])^2, i = 1 .. N);
t21 := t4^2;
t25 := t6^2;
t27 := 1 / (4*t2*t14 - 4*t17*t2 + N*t2 - 4*t21 + 4*t6*t4 - t25);
tmp[1] := (-t1*t2 + t7*t8 + 2*t10*t2)*t27;

```

```

tmp[2]:=((t4 - 1 / 2*t6)*t1 + (2*t14 - 2*t17 + 1 / 2*N)*t8 + t7*t10)*t27;
tmp
end

```

Logarithm

leastsqrs($Y = A + B \ln(t_i)$, t , i , $[A, B]$)

leastsqrs[0]: normal equations

$$-2S_0 + 2AN + 2BT_1 = 0$$

$$-2S_1 + 2AT_1 + 2BT_0 = 0$$

leastsqrs[0]: Solving the normal equations...

leastsqrs[0]: substitution list

$$\left[\Delta = \left(\sum_{i=1}^N \ln(t_i) \right)^2 + \left(\sum_{i=1}^N \ln(t_i)^2 \right) N, S_0 = \sum_{i=1}^N Y_i, T_0 = \sum_{i=1}^N \ln(t_i)^2, T_1 = \sum_{i=1}^N \ln(t_i), S_1 = \sum_{i=1}^N \ln(t_i) Y_i \right]$$

leastsqrs[0]: Verifying the solution...

$$\left[A = \frac{S_0 T_0 - T_1 S_1}{\Delta}, B = \frac{S_1 N - T_1 S_0}{\Delta} \right]$$

eval(% , _subslist)

$$\left[A = \frac{\left(\sum_{i=1}^N Y_i \right) \left(\sum_{i=1}^N \ln(t_i)^2 \right) - \left(\sum_{i=1}^N \ln(t_i) \right) \left(\sum_{i=1}^N \ln(t_i) Y_i \right)}{\left(\sum_{i=1}^N \ln(t_i) \right)^2 + \left(\sum_{i=1}^N \ln(t_i)^2 \right) N}, B = \frac{\left(\sum_{i=1}^N \ln(t_i) Y_i \right) N - \left(\sum_{i=1}^N \ln(t_i) \right) \left(\sum_{i=1}^N Y_i \right)}{\left(\sum_{i=1}^N \ln(t_i) \right)^2 + \left(\sum_{i=1}^N \ln(t_i)^2 \right) N} \right]$$

$tmp := \text{array}(\text{map}(x \rightarrow \text{rhs}(x), \text{eval}(\%, \text{_subslist})))$

$\text{optimize}(\text{makeproc}(tmp, \text{parameters} = [N, Y, t]))$

proc(N, Y, t)

local $tmp, t5, t11, t4, t8, t1, t2;$

$tmp := \text{array}(1 .. 2);$

```

t1 := sum(Y[i], i = 1 .. N);
t2 := sum(ln(t[i])^2, i = 1 .. N);
t4 := sum(ln(t[i]), i = 1 .. N);
t5 := sum(ln(t[i])*Y[i], i = 1 .. N);
t8 := t4^2;
t11 := 1 / (-t8 + t2*N);
tmp[1] := (t2*t1 - t4*t5)*t11;
tmp[2] := (t5*N - t1*t4)*t11;
tmp

```

end

Exponential

leastsqrs $\left(Y = A + B e^{t_i}, t, i, [A, B] \right)$
 leastsqrs[0]: normal equations

$$-2S_0 + 2AN + 2BT_0 = 0$$

$$-2S_1 + 2AT_0 + 2BT_1 = 0$$

leastsqrs[0]: Solving the normal equations...
 leastsqrs[0]: substitution list

$$\begin{aligned} \Delta &= \left(\sum_{i=1}^N e^{t_i} \right)^2 + \left(\sum_{i=1}^N \left(e^{t_i} \right)^2 \right) N, S_0 = \sum_{i=1}^N Y_i, T_0 = \sum_{i=1}^N e^{t_i}, S_1 = \sum_{i=1}^N e^{t_i} Y_i, \\ T_1 &= \sum_{i=1}^N \left(e^{t_i} \right)^2 \end{aligned}$$

leastsqrs[0]: Verifying the solution...

$$A = \frac{S_0 T_1 - T_0 S_1}{\Delta}, B = \frac{S_1 N - T_0 S_0}{\Delta}$$

eval(%,_sublist)

$$A = \frac{\left(\sum_{i=1}^N Y_i \right) \left(\sum_{i=1}^N \left(e^{t_i} \right)^2 \right) - \left(\sum_{i=1}^N e^{t_i} \right) \left(\sum_{i=1}^N e^{t_i} Y_i \right)}{\left(\sum_{i=1}^N e^{t_i} \right)^2 + \left(\sum_{i=1}^N \left(e^{t_i} \right)^2 \right) N},$$

$$B = \frac{\left(\sum_{i=1}^N e^{t_i} Y_i \right) N - \left(\sum_{i=1}^N e^{t_i} \right) \left(\sum_{i=1}^N Y_i \right)}{\left(\sum_{i=1}^N e^{t_i} \right)^2 + \left(\sum_{i=1}^N \left(e^{t_i} \right)^2 \right) N}$$

tmp := array(map(x → rhs(x), eval(%%, _subslist)))

optimize(makeproc(tmp, parameters = [N, Y, t]))

```

proc(N, Y, t)
local t5, t11, t1, t2, tmp, t4, t8;
    tmp := array(1 .. 2);
    t1 := sum(Y[i], i = 1 .. N);
    t2 := sum(exp(t[i])^2, i = 1 .. N);
    t4 := sum(exp(t[i]), i = 1 .. N);
    t5 := sum(exp(t[i])*Y[i], i = 1 .. N);
    t8 := t4^2;
    t11 := 1 / (-t8 + t2*N);
    tmp[1] := (t1*t2 - t4*t5)*t11;
    tmp[2] := (t5*N - t4*t1)*t11;
    tmp
end

```

Exponentially Decaying Sinusoid

The substitution list for this case is quite long, so its output will be suppressed.

printlevel := 1

```

leastsqrs
$$\left( Y = A + e^{-t_i} (B \cos(2 \pi v t_i) + C \sin(2 \pi v t_i)), t, i, [A, B, C] \right)$$

leastsqrs[0]: normal equations

```

$$2 A N + (4 T_0 - 2 T_2) B - 2 S_0 + 4 C T_3 = 0$$

$$(4 T_0 - 2 T_2) A + (8 T_4 - 8 T_6 + 2 T_5) B + (-4 T_8 + 8 T_7) C - 4 S_1 + 2 S_2 = 0$$

$$4 A T_3 + (-4 T_8 + 8 T_7) B - 4 S_3 + 8 C T_1 = 0$$

```

leastsqrs[0]: Solving the normal equations...
leastsqrs[2]: Verifying the solution...

```

soln := %

for p **in** soln **do** print(p) **od**

$$\begin{aligned}
A &= \left(\left(4 T_7^2 + T_8^2 - 4 T_7 T_8 - T_5 T_1 - 4 T_4 T_1 + 4 T_6 T_1 \right) S_0 \right. \\
&\quad \left. + (-4 T_7 T_3 + 4 T_0 T_1 + 2 T_8 T_3 - 2 T_2 T_1) S_1 + (T_2 T_1 + 2 T_7 T_3 - T_8 T_3 - 2 T_0 T_1) S_2 \right. \\
&\quad \left. + (-4 T_3 T_6 + T_3 T_5 + 4 T_3 T_4 - 4 T_0 T_7 + 2 T_7 T_2 + 2 T_0 T_8 - T_2 T_8) S_3 \right) / \Delta \\
B &= \left((-2 T_7 T_3 + 2 T_0 T_1 + T_8 T_3 - T_2 T_1) S_0 + \left(-2 T_1 N + 2 T_3^2 \right) S_1 + \left(T_1 N - T_3^2 \right) S_2 \right. \\
&\quad \left. + ((-T_8 + 2 T_7) N - T_3 (2 T_0 - T_2)) S_3 \right) / \Delta \\
C &= \left(\left(-\frac{1}{2} T_2 T_8 - 2 T_0 T_7 + T_0 T_8 + T_7 T_2 - 2 T_3 T_6 + \frac{1}{2} T_3 T_5 + 2 T_3 T_4 \right) S_0 \right. \\
&\quad \left. + ((-T_8 + 2 T_7) N - T_3 (2 T_0 - T_2)) S_1 + \left(\left(\frac{1}{2} T_8 - T_7 \right) N + \frac{1}{2} T_3 (2 T_0 - T_2) \right) S_2 \right. \\
&\quad \left. + \left(\left(2 T_6 - \frac{1}{2} T_5 - 2 T_4 \right) N + \frac{1}{2} (2 T_0 - T_2)^2 \right) S_3 \right) / \Delta
\end{aligned}$$

The following commands produce an optimized procedure, which we will not show here.

```
tmp := array(map(x → rhs(x), eval(soln, _subslist)))
```

```
optimize(makeproc(tmp, parameters = [N, Y, t]))
```

Sinusoid with Exponentially Decaying Component

$$model := Y = A + B \cos(2 \pi v_1 t_i) + C \sin(2 \pi v_1 t_i) + e^{(-t_i)} (D \cos(2 \pi v_2 t_i) + E \sin(2 \pi v_2 t_i))$$

```
leastsqrs(model, t, i, [A, B, C, D, E])
```

```
leastsqrs[4]: normal equations
```

$$2 N A + (-2 N + 4 T_{11}) B + 4 C T_{12} + (4 T_{13} - 2 T_{15}) D - 2 S_3 + 4 E T_{18} = 0$$

$$\begin{aligned}
&(-2 N + 4 T_{11}) A + (-8 T_{11} + 2 N + 8 T_3) B + (8 T_6 - 4 T_{12}) C \\
&+ (-4 T_{13} - 4 T_9 + 2 T_{15} + 8 T_7) D + (8 T_{10} - 4 T_{18}) E + 2 S_3 - 4 S_5 = 0
\end{aligned}$$

$$4 A T_{12} + (8 T_6 - 4 T_{12}) B + 8 C T_{16} + (8 T_{17} - 4 T_{19}) D - 4 S_4 + 8 E T_8 = 0$$

$$\begin{aligned}
&(4 T_{13} - 2 T_{15}) A + (-4 T_{13} - 4 T_9 + 2 T_{15} + 8 T_7) B + (8 T_{17} - 4 T_{19}) C \\
&+ (8 T_{14} - 8 T_0 + 2 T_1) D + (8 T_2 - 4 T_4) E - 4 S_0 + 2 S_1 = 0
\end{aligned}$$

$$4 A T_{18} + (8 T_{10} - 4 T_{18}) B + 8 C T_8 + (8 T_2 - 4 T_4) D - 4 S_2 + 8 E T_5 = 0$$

```
leastsqrs[4]: Solving the normal equations...
leastsqrs[46]: Verifying the solution...
leastsqrs[569]: Done!
```

soln := %

As we can see, the solution for this model is quite complicated:

cost(soln)

2188 additions + 7749 multiplications + 5 divisions + 5 assignments

map(cost, soln)

*[789 additions + 2864 multiplications + divisions + assignments,
394 additions + 1382 multiplications + divisions + assignments,
394 additions + 1382 multiplications + divisions + assignments,
217 additions + 739 multiplications + divisions + assignments,
394 additions + 1382 multiplications + divisions + assignments]*

cost(_subslst)

254 additions + 1249 multiplications + 1108 sums + 27 assignments

map(cost, eval(soln, _subslst))

*[1043 additions + 4113 multiplications + 3968 sums + divisions + assignments,
648 additions + 2631 multiplications + 2487 sums + divisions + assignments,
648 additions + 2631 multiplications + 2487 sums + divisions + assignments,
471 additions + 1988 multiplications + 1842 sums + divisions + assignments,
648 additions + 2631 multiplications + 2487 sums + divisions + assignments]*

However, it optimizes well:

optsoln := optimize(soln, tryhard)

cost(optsoln)

739 additions + 1061 multiplications + divisions + 26 subscripts + 447 assignments

The following command produces an optimized procedure, which we will not show here.

makeproc(eval([optsoln], _subslst), parameters = [N, Y, t], globals = [A, B, C, D, E])